// 4_Merge 2010-2011-2013
// //////////////////////////////////// 

// This do-file merges previously processed databases. 

// ////////////////////////////////////////////////////////////////////////
// INDEX
// I.	Setup of variables 2010
// II.	Merge of variables 2010 with variables 2011
// III.	Setup of variables 2011
// IV.	Setup of variables 2013
// V.	Merge of variables 2013 with variables 2010-2011
// ////////////////////////////////////////////////////////////////////////



clear all
set more off

capture log close

log using "${path_logs}/4_Merge 2010-2011-2013.log", replace


// ////////////////////////////////////////////////////////////////////////
// I.	Setup of variables 2010
// ////////////////////////////////////////////////////////////////////////

use "${path_data}/variables2010.dta", clear

foreach x in  mesp mesp_p C anymesp TT id dependiente independiente trabaja ylaborales numper nucleos sexo edad basica media_inc media superior ydeptes yindeptes fichaps {
	rename `x' `x'2010	
}

sort id2010

tempname variables2010 
save "${path_data}/`variables2010'", replace


// ////////////////////////////////////////////////////////////////////////
// II.	Merge of variables 2010 with variables 2011
// ////////////////////////////////////////////////////////////////////////

use "${path_data}/variables2011",clear
sort id2010

merge id2010 using "${path_data}/`variables2010'"

drop _merge mesp2010 mesp_p2010 C2010 anymesp2010 TT2010

// Updating infomation of Social Security Card score for those not interviewed in 2010
replace fichaps=fichaps2010 if ind_c2011==.



save "${path_data}/variables2010_2011",replace


// ////////////////////////////////////////////////////////////////////////
// III.	Setup of variables 2011
// ////////////////////////////////////////////////////////////////////////

foreach x in id dependiente independiente trabaja ylaborales sexo edad basica media_inc media superior ydeptes yindeptes fichaps ht marketing stock cost plan tot_pr dinero_mano ver_inven ver_registro {
	rename `x' `x'2011	
}

save "${path_data}/variables2010_2011",replace

// ////////////////////////////////////////////////////////////////////////
// IV.	Setup of variables 2013
// ////////////////////////////////////////////////////////////////////////

use "${path_data}/variables2013.dta", clear

foreach x in id dependiente independiente trabaja ylaborales sexo edad ydeptes yindeptes marketing stock cost plan tot_pr dinero_mano ver_inven ver_registro totalactivos ht {
	rename `x' `x'2013	
}

sort id2010

tempname variables2013
save "${path_data}/`variables2013'", replace

// ////////////////////////////////////////////////////////////////////////
// V.	Merge of variables 2013 with variables 2010-2011
// ////////////////////////////////////////////////////////////////////////

use "${path_data}/variables2010_2011",clear
sort id2010

merge id2010 using "${path_data}/`variables2013'"
drop _merge

save "${path_data}/variables2010_2011_2013",replace


erase "${path_data}/`variables2010'.dta"
erase "${path_data}/`variables2013'.dta"

log close
